
Abstract
Since its 3.3 release, Modelica offers the possibility to
specify models of dynamical systems with multiple modes
having different DAE-based dynamics. However, the han-
dling of such models by the current Modelica tools is not
satisfactory, with mathematically sound models yielding
exceptions at runtime. In this article, we propose a system-
atic way of rewriting a multimode Modelica model, based
on the results of an already implemented multimode struc-
tural analysis. The rewritten Modelica model is guaranteed
to be correctly compiled by state-of-the-art Modelica tools.
Simulation results are presented on a simple, yet meaning-
ful, physical system whose original Modelica model is not
correctly handled by state-of-the-art Modelica tools.
Keywords: Modelica, multimode DAE, structural analysis,
model transformations

1 Introduction
Since version 3.3, the Modelica language offers the pos-
sibility of specifying multimode dynamics, by describing
state machines with different DAE dynamics in each dif-
ferent state (Elmqvist et al., 2012). This feature enables
describing large complex cyber-physical systems with dif-
ferent behaviors in different modes.

While being undoubtedly valuable, multimode modeling
has been the source of serious difficulties for non-expert
users of the current generation of Modelica tools. Indeed,
while many large-scale Modelica models are properly han-
dled, some physically meaningful models do not result in
correct simulations with most Modelica tools. It is actually
not difficult to construct such problematic models, thus,
chances are significant to produce such bad cases in large
models. Quite often, end users have to ask Modelica ex-
perts, or even tool developers themselves, to tweak their
models in order to make them work as expected. This situ-
ation hinders a wider spreading of Modelica tools among a
larger class of users, such as Simulink-trained engineers.

«««< .mine New language constructs have been pro-
posed in the past to address the limited capability of the
Modelica language to handle multimode models. The
Sol (Zimmer, 2010) and the Functional Hybrid Model-
ing (Nilsson and Giorgidze, 2010) languages have been
designed with the capability to enable and disable equa-
tions, depending on the current mode of the system. For

Some years ago, we started a project aiming at address-
ing all the above issues, with a different perspective in
mind, that consists in privileging compile-time, rather than
runtime, analyses. In (Benveniste et al., 2019, 2020) we
explain our approach, and we illustrate it on two simple,
yet physically meaningful, examples in (Benveniste et al.,
2021). One key feature of this approach is structural analy-
sis: it is important that this task is performed for each mode
and each mode change at compile time, in order to avoid
unexpected behaviour at runtime. In (Caillaud et al., 2020),
we present an effective approach to achieve compile-time,
mode-dependent, structural analysis without enumerating
the modes (as this would not be able to scale up).

The advantages we see in our approach are twofold:
(i) it provides, at compile-time, invaluable information
that helps users debug their models, and (ii) efficient code
generation is possible since the automatic differentiation
of latent equations can be done at compile-time and blocks
of equations can be compiled into functions that can be
passed directly to numerical solvers, without any further
processing.

In this article, we demonstrate how the results of this
multimode structural analysis can be used for transforming
a multimode Modelica model into its RIMIS (Reduced



Index Mode-Independent Structure) form, which is guaran-
teed to yield correct execution on state-of-the-art Modelica
tools. This method is illustrated on a water tank model for
which current Modelica tools fail to execute; in this model,
the differentiation index depends on the mode, which is
a problem for these tools. In particular, we explain how
existing structural analysis methods fail to yield correct
execution code for this model, then demonstrate the gen-
eration of a target code under RIMIS form, resulting in a
correct simulation of the model. Our approach is then for-
malized for its broad application to problematic multimode
models.

2 The Water Tank system and failed
simulations with Modelica tools

The Water Tank system is a simple model of a closed tank
with a variable water inflow z and a default outflow y0,
where water is considered incompressible. When the tank
is full, a positive flow correction yh is added to the outflow,
as the tank cannot store more water; conversely, when the
tank is empty, a negative flow correction yl is added to the
outflow.

The corresponding Modelica model, given in Figure 1,
uses two complementarity conditions (van der Schaft and
Schumacher, 1998) for the flow corrections. The first one,
encoded by the multimode equations eh1 and eh2, de-
pends on the Boolean variable bh, which is true if and
only if variable sh is nonnegative. The combined effect
of these two equations is that xmax−x and yh are always
nonnegative, and that at least one of those is equal to 0
at any time. Equations el1 and el2 encode the second
complementarity condition in a similar way.

This model fails to simulate properly with both Open-
Modelica 1.17.0 (Fritzson et al., 2020) and Dymola
2021 (Dassault Systèmes AB); Figure 2 shows the out-
put of Dymola 2021. The root cause is that state-of-the-art
Modelica tools perform an approximate structural analy-
sis, disregarding the fact that the structure of the system is
mode-dependent. A more detailed explanation is provided
in Section 3.1.

3 Structural analysis: from single- to
multi-mode

DAE-based languages and tools rely on structural anal-
ysis as a required preprocessing step of a DAE system,
needed for the generation of simulation code. This analysis
turns the original system into a reduced index (Campbell
and Gear, 1995) system, amenable to numerical solvers,
by differentiating one or several times all or part of the
equations.

Well-understood methods such as the renowned Pan-
telides algorithm (Pantelides, 1988), the dummy derivatives
method (Mattsson and Söderlind, 1993) or the less known
Σ-method (Pryce, 2001) can be used for single-mode DAE
systems; however, the structural analysis of multimode

model WaterTank
  Real t(start=0,fixed=true); // time (to define input flow)
  constant Real xmax = 1.0; // max water quantity
  constant Real xmin = 0.0; // min water quatity
  constant Real y0 = 6.667; // default output flow 
  constant Real rho = 0.8; // input flow parameter
  Real x(start=0.5,fixed=true); // stored water mass
  Real yh; // output flow correction, when tank is full
  Real yl; // output flow correction, when tank is empty
  Real z; // input flow
  Real sh; // parameter of the full-tank CC
  Real sl; // parameter of the empty-tank CC
  Boolean bh(start=false,fixed=true); // mode full-tank
  Boolean bl(start=false,fixed=true); // mode empty-tank
  // bh and bl satisfy assertion not (bh and bl)
equation
  // input flow law
  /* et: */ der(t)=1;
  /* e1: */ z = rho*y0*(1+
            Modelica.Math.cos(2*Modelica.Constants.pi*t));
  // tank level differential equation
  /* e2: */ der(x) = z + yl - yh - y0;
  // Complementarity condition 0 <= xmax - x # yh >= 0
  bh = (sh >= 0);
  /* eh1: */ sh = if bh then yh else x - xmax;
  /* eh2: */ 0 = if bh then x - xmax else yh;
  // complementarity condition 0 <= x - xmin # yl >= 0
  bl = (sl >= 0);
  /* el1: */ sl = if bl then yl else xmin - x;
  /* el2: */ 0 = if bl then xmin - x else yl;
end WaterTank;

Figure 1. Modelica model of the Water Tank system. Comments
of the form /* id: */ define equation labels appearing in
the dependency graphs in Figures 3 and 4.

DAE systems is still in its infancy, and even state-of-the-art
Modelica tools have to rely, at least in part, on an approxi-
mate ‘single-mode’ structural analysis for the generation
of simulation code from multimode models.

We show how the use of such single-mode methods
can lead to the runtime errors observed on the Water Tank
model shown above. We then introduce the exact multi-
mode structural analysis performed by the IsamDAE tool,
which will be used for the transformation of multimode
models at the core of this article.

3.1 Approximate structural analysis
Structural analysis of a DAE system only relies on the
knowledge of which numerical variables appear in which
equations. As such, an approximate structural analysis of

Figure 2. Simulation of the Water Tank system with Dymola
2021, failing with a division by zero exception.



x -- el2 -> yl x yl -- el1 -> sl

yh yl z -- e2 -> x'

x -- eh2 -> yh x yh -- eh1 -> sh

t -- e1 -> z

et -> t'

Figure 3. Dependency graph resulting from the approximate
structural analysis of the Water Tank model. Vertices are equa-
tion blocks of the form R−E →W , where: E is the block of
equations; R is a set of variables to read (they are free variables,
i.e., parameters of the block of equations); and W is a set of vari-
ables to write (they are the unknowns of the block of equations).
When R is empty, the shorthand notation is E →W . Edges ex-
press causal dependencies, meaning that a block can be solved
only after all its predecessors have been solved.

a multimode DAE system can be performed by abstract-
ing away all mode dependencies inside the equations; for
instance, an equation x = if cond then y else z will
be regarded by the approximate structural analysis as an
equation involving variables x, y and z.

Such an analysis of the Water Tank model shown in
Figure 1 results in the decomposition shown in Figure 3.
In this decomposition, equation eh2 has to be solved for
the variable yh.

When performing the pivoting of this equation, mode
dependencies have to be taken into account again. Equation
eh2 reads:

0 = if bh then x−xmax else yh

which can be rewritten as an equation of the form 0 =
a yh+b where a and b are mode-dependent:

0 =(if bh then 0 else 1)×yh
+(if bh then x−xmax else 0)

Unknown yh can finally be isolated:

yh=−if bh then x−xmax else 0
if bh then 0 else 1

(1)

This technique may be used for the generation of simu-
lation code, but in this case, a problem is bound to occur
when Boolean variable bh is true. As a matter of fact,
equation (1) is exactly the equation responsible for the di-
vision by zero exception shown in Figure 2, which occurs
at the initial time, when bh is true.

not bl:

not bh and not bl:
yh yl z -- e2 -> x'

not bh and not bl

x' yl z -- e2 -> yh

Figure 4. Conditional Dependency Graph resulting from the
multimode structural analysis of the Water Tank model. Vertices
are conditional equation blocks of the form p : R−E→W , where:
E is the block of equations; p is a Boolean condition, defining
the set of modes in which the block has to be solved; R is a set of
variables to read, or free variables, i.e., parameters of the block of
equations; and W is a set of variables to write, meaning that they
are the unknowns of the block of equations. When R is empty,
the shorthand notation is p : E→W . When p is the proposition
true, it is omitted, and the notation becomes: R−E →W , or
E →W . Edges express causal dependencies, meaning that a
block can be solved only after all its predecessors have been
solved. They are labeled by Boolean conditions, characterizing
the modes in which the dependency applies.

3.2 Exact multimode structural analysis

The IsamDAE1 tool (Caillaud et al., 2020) has been used
to perform a multimode structural analysis of the model,
resulting in the Conditional Dependency Graph (CDG)
shown in Figure 4.

Remark that the differentiation index of the system is
mode-dependent. For instance, equation el2 is used differ-
entiated, to compute the derivative of x, when bl is true,
while it is kept undifferentiated, to compute yl, when bl
is false. Also notice that equation eh2 is no longer used
to compute yh in all modes, but only when bh is false,
thus preventing the runtime error explained above.

We shall see next how the CDG (Figure 4) can be used
to transform the model into an equivalent one, that triggers
no runtime error when using Modelica tools based on an
approximate structural analysis.

1https://team.inria.fr/hycomes/software/isamdae/

https://team.inria.fr/hycomes/software/isamdae/


4 A Reduced Index Mode-Independent
Structure (RIMIS) form

Using multimode structural analysis to transform a mul-
timode Modelica model into a reduced-index model, that
simulates correctly with state-of-the-art Modelica tools, is
made difficult by the fact that the Modelica language does
not permit to enable or disable an equation depending on
the mode. Based on this limitation, the basic principle of
our model transformation is to evaluate all equation blocks
of the CDG in a mode-independent fashion, irrespectively
of the mode in which the system is. Of course, this leads
to useless computations during simulation. However, this
turns out to be a systematic way to ensure a correct simula-
tion of multimode Modelica models.

The method proposed in this paper is detailed below, in
informal terms, then illustrated on a simple example. A
mathematical definition of the transformation is detailed
in Section 6. Remark that models with initial equations,
when or reinit statements are not covered in this pa-
per. Also note that models with non-scalar variables or
class instances of any kind are not considered here. It is
assumed that the models have been flattened according
to the procedure described in Chapter 5 of the Modelica
Language Specification (Modelica Association, 2021). Be-
cause of a current restriction of the IsamDAE software,
mode variables are assumed to be of type Boolean.

4.1 The RIMIS form transformation
The method decomposes in the following seven steps:

1. Conditional Dependency Graph: The CDG of the
source model is computed by the multimode struc-
tural analysis method. This graph defines a block-
triangular decomposition of the reduced-index sys-
tem, for each mode of the system. It will be used
throughout the transformation.

2. Source Variables: Variable declarations are copied
unchanged, with the exception of real variables,
whose initialization parts are removed.

3. Replicate and Dummy Derivative Variables: For
each block of the CDG, replicates of written vari-
ables (unknowns) are declared. Whenever an un-
known appears differentiated, a dummy derivative
variable (Mattsson and Söderlind, 1993) is declared.
Initialization statements for state variables are copied
from the source model. As an optional optimization,
non-leading replicate variables can be factored among
a disjunction of modes, in order to decrease the num-
ber of variables in the resulting model.

4. Mode Equations: Equations defining mode variables
are copied unchanged. For the sake of simplicity,
these equations are assumed to be of the form b =
(expr >= 0), where expr is a real expression.

5. Replicate and Dummy Equations: Equations are
replaced with replicates, according to the following
principle:

For each block in the CDG, equations appearing in
this block are replicated, substituting (i) every writ-
ten variable (unknown of the block) by the replicate
declared in step 3, and (ii) every read variable (pa-
rameter of the block) by the corresponding replicate,
if it is a leading variable. Both mode variables and
read state variables are left unchanged.

As a result, the single-mode structural analysis of the
resulting equation system yields a block-triangular
decomposition that contains all the blocks of the CDG
obtained by the multimode structural analysis of the
original model.

For each equation in the fresh model, the proposi-
tional formula conditioning the block in which this
equation appears can be taken into account: a partial
evaluation of the equation is performed (Jones et al.,
1993). This has the effect of simplifying the equation,
by eliminating some of the conditionals (if ... then
... else ... operators).

Note that the resulting equations may still be multi-
mode: in general, not all conditionals can be elim-
inated by partial evaluation. However, the fact that
the structure of the resulting equations is indepen-
dent of the mode is still guaranteed: the multimode
structural analysis ensures that each equation block
has the same structure (in particular, the same read
and written variables) in all the modes in which it is
defined, even if one or several of its equations contain
conditional statements.

First-order differential equations are also added in
accordance to the dummy derivatives method.

6. Multiplexing Equations: In order to retrieve the val-
ues of the source model variables from the replicates
in the fresh model, mutiplexing equations have to
be added. These are multimode equations, contain-
ing conditional operators, but these equations contain
no dynamics: each multiplexing equation focuses on
a source model variable that corresponds to several
replicates in the transformed model, specifying which
of the latter currently holds the value of the former.

7. Reinitializations: Reinitialization statements finally
have to be inserted, in order to reset replicate variables
that are state variables to a correct value upon the
occurrence of a mode switching. Therefore, these
statements are triggered by mode changes.

4.2 Transformation of a simple model
We illustrate the method on a simplistic, yet relevant, two
equations model:



p: e -> x

not p: e -> x'

Figure 5. CDG of the Two Equations model.

Figure 6. Failed simulation of the Two Equations model with
Dymola 2021.

model TwoEquations
  Real x(start=0,fixed=true);
  Boolean p(start=false,fixed=true);
equation
  p = (x >= 1);
  1 = if p then x else der(x);
end TwoEquations;

This model has one real equation, one Boolean equation,
and no particular physical meaning. However, it captures in
a nutshell the difficulty raised with the Water Tank system.
As a matter of fact, the CDG (Figure 5) resulting from the
multimode structural analysis distinguishes between two
cases:

• when p is true, x is a leading variable, meaning that
it is the unknown that needs to be solved;

• when p is false, the leading variable is x′, the first-
order time derivative of x, while x itself is a state
variable.

The approximate structural analysis of both Dymola and
OpenModelica determines that the leading variable is x′

in all modes; however, the real equation is singular in x′

when p is true. Unsurprisingly, an exception is raised
during simulation, as shown in Figure 6.

Let us apply the transformation one step after the other:

1. The CDG graph of the source model is shown in
Figure 5.

2. Declarations of variables x and p are copied.model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

Remark that the declaration of x has been stripped of
its initialization part.

3. Replicate variables are created according to the two
blocks of the CDG. Two leading replicate variables
x_2 (holding the value of x if p holds) and x_p_3
(holding the value of x′ if not p holds), and one
state replicate variable x_3 that is meaningful only if
not p holds, are declared.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

Note that the initialization of variable x in the source
model is copied here, to initialize the replicate state
variable x_3.

4. One mode equation is copied from the source model.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

5. Replicate equations are generated from the CDG,
which has two blocks of one equation each.

From the block p : e→ x, one replicate equation is
generated by replacing variable x with its replicate
x_2, then performing the partial evaluation (Jones
et al., 1993) under the assumption that the Boolean
condition p holds.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

From the second block not p : e→ x′, one replicate
equation is generated in a similar way.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

A differential equation is also generated, linking repli-
cate variable x_3 with its dummy derivative x_p_3.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

6. One multiplexing equation is generated, to be
solved for variable x.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

7. Finally, the only case in which a state variable has to
be reinitialized is when entering the mode not p.
The value of replicate variable x_3 is then set to be
the left limit of x.

model OneEquation_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end OneEquation_rimis;

The complete RIMIS form of the Two Equations model
is given in Figure 7. The result of the successful simulation
of this model is shown in Figure 8. Remark that the mode
switching from p = false to p = true is correct, and
that the reinitialization statement is never evaluated, as p
remains true forever after time t= 1.



model TwoEquations_rimis
  // Source variables
  Real x;
  Boolean p(start=false,fixed=true);
  // Replicate variables
  Real x_2;
  Real x_p_3;
  Real x_3(start=0,fixed=true);
equation
  // Mode equations
  p = (x >= 1);
  // Differential equations
  der(x_3) = x_p_3;
  // Multiplexing
  x = if p then x_2 else x_3;
  // Block e_3 -> x_p_3
  /* e_3 : */ 1 = x_p_3;
  // Block e_2 -> x_2
  /* e_2 : */ 1 = x_2;
  // Replicate reinitializations
  when not p then
    reinit(x_3,pre(x));
  end when;
end TwoEquations_rimis;

Figure 7. Two Equations model in RIMIS form.

Figure 8. Simulation of the Two Equations model in RIMIS
form with Dymola 2021.

5 Successful simulations of the Water
Tank system in RIMIS form

The RIMIS transformation is illustrated on the Water Tank
model (Figure 1); the resulting model is shown in Figure 9.
Simulation results obtained with Dymola 2021 are shown
in Figure 10. It can be seen that the simulation is successful,
with a correct behavior of the Water Tank system, while the
simulation of the original model failed (Figure 2). A cor-
rect simulation has also been obtained with OpenModelica
1.17.0 (Fritzson et al., 2020), under the provision that the
Newton solver is used instead of the KINSOL nonlinear
solver.

6 Formalizing the RIMIS form trans-
formation

The mathematical definition of the RIMIS form transfor-
mation relies on the partial evaluation of equations. Once
variable renaming is also properly defined, the seven-step
transformation mentioned in Section 4.1 is formalized. Fi-
nally, an optimization aiming at reducing the transformed
model is presented.

6.1 Partial evaluation of expressions and equa-
tions

Partial evaluation is an umbrella name for a set of program
transformation techniques that aim at specializing a pro-
gram by taking into account prior knowledge on its input
data, possibly improving its performances (Jones et al.,
1993; Danvy et al., 1996).

In the context of the Modelica language, consider a
Boolean expression q, and a real expression e. The partial
evaluation of expression e, assuming q, is an expression
e′= πq(e), such that q implies e= e′ and free(e′)⊆ free(e),
where free(.) is the set of free variables appearing in an
expression.

To define the partial evaluation operator π , and for the
sake of clarity, we only consider the subset of the Modelica
expression language defined by the following grammar,
where p is a Modelica Boolean expression:

e ::= c where c is a constant
| e op e where op ∈ {+,-,*, . . .}
| v where v is an identifier
| v(e, . . .e)
| if p then e else e

Given a Boolean expression q and a real expression
e, the partial evaluation of e, assuming q, is defined by
induction on the structure of e:

πq(c) ≡ c
πq(e1 op e2) ≡ πq(e1) op πq(e2)
πq(v) ≡ v
πq(v(e1, . . .en)) ≡ v(πq(e1), . . .πq(en))
πq(if p then eT else eF) ≡ condq(p,eT ,eF)

where

condq(p,eT ,eF) ≡∣∣∣∣∣∣∣∣∣∣∣∣∣

πq and p(eT ) if q and not p
is unsatisfiable, else

πq and not p(eF) if q and p
is unsatisfiable, else

if r where r is such that:
then πq and p(eT ) p and q implies r, and
else πq and not p(eF) r implies p or not q

In the above definition, condition r is not unique: whenever
possible, it should be chosen such that it is more concise
than p.



model WaterTankRIMIS
  // Constants
  constant Real xmax = 1.0;
  constant Real xmin = 0.0;
  constant Real y0 = 6.667;
  constant Real rho = 0.8;
  // Variables
  Real t(start=0,fixed=true);
  Real x(start=0.5,fixed=true);
  Real yh;
  Real yl;
  Real z;
  Real sh;
  Real sl;
  Boolean bh(start=false,fixed=true);
  Boolean bl(start=false,fixed=true);
  // Dummy derivatives
  Real t_p;
  Real x_p;
  // Replicated algebraic variables
  Real sh_5; // sh if not bh
  Real sh_6; // sh if bh
  Real sl_2; // sl if not bl
  Real sl_4; // sl if bl
  Real x_p_4; // x' if bl
  Real x_p_7; // x' if not bh and not bl
  Real x_p_6; // x' if bh
  Real yh_5; // yh if not bh
  Real yh_6; // yh if bh
  Real yl_2; // yl if not bl
  Real yl_4; // yl if bl
equation
  // Boolean equations
  bh = (sh >= 0);
  bl = (sl >= 0);

  // Differential equations
  der(t) = t_p;
  der(x) = x_p;
  // Multiplexing equations
  yh = if bh then yh_6 else yh_5;
  yl = if bl then yl_4 else yl_2;
  sh = if bh then sh_6 else sh_5;
  sl = if bl then sl_4 else sl_2;
  x_p = if bh then x_p_6 else
          if bl then x_p_4 else x_p_7;
  // Block et -> t'
  t_p = 1;
  // Block not bh: x -- eh1 -> sh
  sh_5 = x - xmax;
  // Block not bl: x -- el1 -> sl
  sl_2 = xmin - x;
  // Block bl: el2' -> x'
  x_p_4 = 0;
  // Block not bh: eh2 -> yh
  yh_5 = 0;
  // Block x -- e1 -> z
  z = rho*y0*(1+
      Modelica.Math.cos(2*Modelica.Constants.pi*t));
  // Block not bl: el2 -> yl
  yl_2 = 0;
  // Block bh: eh2' -> x'
  x_p_6 = 0;
  // Block bl: x' yh z -- e2 -> yl
  yl_4 = y0 + x_p_4 + yh_5 - z;
  // Block not bh & not bl: yh yl z -- e2 -> x'
  x_p_7 = z + yl_2 - yh_5 - y0;
  // Block bh: x' yl z  -- e2 -> yh
  yh_6 = z + yl_2 - x_p_6 - y0;
  // Block bl: yl -- el1 -> sl
  sl_4 = yl_4;
  // Block bh: yh -- eh1 -> sh
  sh_6 = yh_6;
end WaterTankRIMIS;

Figure 9. The Water Tank system in RIMIS form.

The extension of the partial evaluation operator to equa-
tions is straightforward:

πq(eLHS = eRHS) ≡ πq(eLHS) = πq(eRHS) .

6.2 Variable renaming
Before moving to the formal definition of the RIMIS trans-
formation, variable renaming must be defined, in order to
declare replicate variables and transform equations into
their replicates.

Given a Boolean expression p, an identifier v, and a
differentiation order n≥ 0, the replicate of the n-th order
derivative of v, under condition p, is the identifier ρn

p(v).
The operator ρ is assumed to satisfy the following axioms:

(Identity) ρ0
true(u) = u

(Injectivity) ρn
p(u) = ρm

q (v) implies u = v and
p ⇐⇒ q and
n = m

Checking the equivalence of two Boolean expressions is,
in general, a difficult problem. In this article, Boolean

expressions that appear in conditional statements are re-
stricted to propositional formulas only. Mode equations
are restricted to the form v = (e >= 0), where e is an
affine expression. Under these assumptions, equivalence
checking can be done with BDDAPRON, a logico-numerical
abstract domain library (Jeannet, 2012) combining BDDs
(Boolean Decision Diagrams) (Bryant, 1986) and poly-
hedra (Schrijver, 1998). Such a use of BDDAPRON is
considered, among other program analyses, in Chapter 7
of (Schrammel, 2012).

6.3 Formal definition of the RIMIS form
transformation

Consider a Modelica model M that can be decomposed in
the following parts:

M ≡ MD]RD]RI]ME]RE

where:

• MD is the set of mode (Boolean) variable declarations
and initializations;



Figure 10. Simulation of the Water Tank system in RIMIS form
with Dymola 2021.

• RD is the set of real variable declarations, stripped of
their initializations;

• RI is the set of real variable initializations;

• ME is the set of mode variable equations;

• RE is the set of real equations.

Remark that models with when and reinit statements
are not covered by the RIMIS form transformation, as this
would require a multimode structural analysis of mode
changes (Benveniste et al., 2020), that is not yet imple-
mented in the IsamDAE software (Caillaud et al., 2020).
Because of a current restriction of IsamDAE, mode vari-
ables are assumed to be Boolean.

Model M is assumed to be structurally nonsingular in
all modes. Its CDG computed by the multimode structural
analysis (Caillaud et al., 2020) consists in a set of blocks of
equations and a set of directed edges between blocks; let
Blocks and Edges denote the corresponding sets. A block
b ∈ Blocks consists of four parts:

• cond(b), a Boolean expression;

• Eqs(b), a set of equations, possibly differentiated;

• Read(b), a set of read variables (parameters of the
block of equations);

• Write(b), a set of written variables (unknowns of the
block of equations).

Elements of Eqs(b) are pairs of the form (0 = e,k), where
e is an expression and k ≥ 0 is a differentiation order. Ele-
ments of Read(b) and Write(b) are pairs of the form (u,k),
where u is an identifier and k ≥ 0 is a differentiation order.
An edge g ∈ Edges consists of three parts:

• cond(g), a Boolean expression;

• from(g), to(g) ∈ Blocks, two blocks.

The meaning of an edge g is that whenever cond(g) holds,
block from(g) has to be solved before block to(g). By
construction, cond(g) implies both cond(from(g)) and
cond(to(g)).

In addition, the multimode structural analysis computes
several functions and predicates on (differentiated) vari-
ables v = (u,k):

• leadingp(v) decides whether variable u is a leading
variable in some mode satisfying the Boolean formula
p;

• algebraicp(v) decides whether u is an algebraic vari-
able in some mode satisfying p;

• statep(v) decides whether u is a state variable in some
mode satisfying p.

For the sake of clarity, the following nota-
tions are introduced: leading(b) = {v ∈ Read(b) ∪
Write(b)| leadingcond(b)(v)} is the set of leading variables
appearing in block b; Defp(v) is the set of blocks that
define variable v in some mode satisfying the Boolean
formula p, either because v itself is written, or because a
higher order derivative of it is written:

Defp(u,k) = {b ∈ Blocks | p∧ cond(b) is satisfiable,
and ∃k′ ≥ k, (u,k′) ∈Write(b)}

The resulting RIMIS form model can be decomposed in
several parts:

RIMIS ≡ MD]RD]DECL] INIT]
ME]REPL]MULTI]DIFF]REINIT

where:

• MD is the set of mode (Boolean) variable declarations
and initializations, taken from M;

• RD is the set of real variable declarations, taken from
M;

• DECL is the set of replicate variable declarations,
defined below;

• INIT is the set of replicate variable initializations,
defined below;

• ME is the set of mode variable equations, taken from
M;

• REPL is the set of replicate equations, defined below;

• MULTI is the set of multiplexing equations, defined
below;

• DIFF is the set of differential equations, defined be-
low;

• REINIT is the set of reinitialization equations, de-
fined below.



Replicate variable declarations (Section 4.1, step 3)
consist in the declaration of the following set of real vari-
ables:

DECL ≡
⋃

b∈Blocks,(u,k)∈Read(b)∪Write(b){
ρ i

cond(b)(u) | 0≤ i≤ k
} .

Replicate variable initializations (Section 4.1, step 3)
consist in the initialization of all replicate variables
ρ0

cond(b)(u) that are state variables, with the initialization
expression for u in M (RI(u)):

INIT ≡
{
(ρ0

p(u),RI(u))|ρ0
p(u) ∈ DECL and statep(u,0)

}
where ρ is a fixed replication operator as defined in Sec-
tion 6.2.

Replicate equations (Section 4.1, step 5) consist in the
differentiation to a given order of the equations of each
block of equations:

REPL ≡
⋃

b∈Blocks{
σb(πcond(b)(δk(q))) | (q,k) ∈ Eqs(b)

}
where π is the partial evaluation operator defined in Sec-
tion 6.1, equation δk(q) is the k-th order differentiation of
equation q, and σb is the substitution operator such that
σb(q) substitutes any variable u in equation q with the repli-
cate variable ρ0

cond(b)(u), any derivative of the form der(u)

by the replicate variable ρ1
cond(b)(u), and so on for higher

order derivatives.
Multiplexing equations (Section 4.1, step 6) serve two

purposes: (i) linking written variables and read variables in
different blocks, and (ii) defining the original real variables
from M:
MULTI =

⋃
b∈Blocks,v=(u,k)∈Read(b)
{ρk

cond(b)(u) = casev(Defcond(b)(v))} ∪⋃
u∈RD{u = caseu,0(Deftrue(u,0)}

where casev is defined by induction over the set of blocks
Deftrue(v) that define variable v in some mode:

case(u,k)({b}) = ρk
cond(b)(u)

casev=(u,k)(b]B) = if cond(b)
then ρk

cond(b)(u)
else casev(B)

Differential equations (Section 4.1, step 5) serve the
purpose of defining replicate state variables from the repli-
cate dummy derivatives:

DIFF =
⋃

b∈Blocks,(u,k)∈Write(b)
{der(ρ i

cond(b)(u)) = ρ
i+1
cond(b)(u)}0≤i≤k−1

Finally, upon the occurrence of a mode change, reinitial-
ization statements (Section 4.1, step 7) serve the purpose
of copying the state vector from a formerly active replicate
state variable to a newly active one:

REINIT =
⋃

b∈Blocks,(u,1)∈Write(b)
{when cond(b) then
reinit( ρ0

cond(b)(u) , pre(u));
endwhen}

6.4 Optimization
Modelica code generated with the procedure described
in Section 6.3 may contain multiplexing equations and
reinitialization statements that can be eliminated thanks to
the optimization described below.

It may happen that a multiplexing equation is of the
form ρk

p(u) = ρk
p′(u). This typically happens when a block

b ∈ Blocks reads a variable that is written by exactly one
block b′ ∈ Blocks. In this case, no multiplexing equation
needs to be generated, and replicate variable ρk

p(u) does
not need to be declared. Instead, every occurrence of ρk

p(u)
in equations q ∈ Eqs(b) shall be replaced by ρk

p′(u).
Remark that this optimization has been applied to the

Water Tank model in RIMIS form (Figure 9). For instance,
equation sh_5= x−xmax refers directly to variable x in-
stead of variable x_5, sparing both the declaration of the
replicate variable x_5 and the generation of the multiplex-
ing equation x = x_5. The same optimization has been
applied to variable z.

7 Conclusion
We presented a method for transforming multimode Model-
ica models that yield simulation errors with state-of-the-art
Modelica tools (such as Dymola 2021 and OpenModelica
1.17.0) into Reduced Index Mode-Independent Structure
(RIMIS) models that simulate correctly with the same tools.

This model transformation relies on the multimode struc-
tural analysis as performed by the IsamDAE tool (Caillaud
et al., 2020). The output of this structural analysis, which
is a Conditional Dependency Graph (CDG) describing all
possible equation blocks in all modes and their dependen-
cies, is used to replicate equations and real variables as
needed. This is performed in such a way that the approx-
imate structural analysis implemented in most Modelica
tools will create the same equation blocks. Dummy deriva-
tives (Mattsson and Söderlind, 1993) are also used so that
the resulting model is of index 0.

The 7-step RIMIS transformation was detailed on a very
simple multimode model, then applied to the Modelica
model of a water tank system; we showed that, while both
source models cause division by zero errors at runtime,
their RIMIS forms simulate correctly with both Dymola
2021 and OpenModelica 1.17.0, yielding the expected be-
haviors for their variables. This process was formalized,
paving the way towards its automation for the handling
of a wider class of multimode models by state-of-the-art
Modelica tools.

A possible drawback of this approach is that the size of
the RIMIS model may a priori be exponential in the size
of the source model, as both equations and real variables
could be replicated once for every mode of the system.
However, experiments on a number of parametric models
with the IsamDAE tool show that the number of blocks
in the CDG of such models tend to be linear in their size,
except for rare pathological cases. As such, the size of the
RIMIS form of a multimode Modelica model will, in a vast



majority of cases, be linear in the size of the original model,
thus making our approach tractable even for large models.

As a concluding remark, it can be noted that the illustra-
tive models in this article are only made of linear equations,
so that the evaluation of all equation blocks, both active and
inactive, at every time step is not an issue. For nonlinear
blocks, not only could this approach be computationally
expensive, but it might fail altogether, as such blocks might
be singular outside of a given subset of the modes.

A simple fix, that was not detailed above, consists in
transforming the equations from such blocks into condi-
tional equations, so that they become trivial equations out-
side of the set of modes in which they have to be considered.
The matching between equations and variables that is com-
puted during the multimode structural analysis can be used
for this task, as it basically tells ‘which variable has to be
solved using which equation’; a nonlinear equation could
then be replaced with the simple assignment of a default
value to its matched real variable in the modes in which the
equation block is inactive. This additional transformation
would still preserve the structure of the model, in the sense
that the approximate structural analysis would still result
in solving the same blocks for the same real variables.

